

Сетевые и принтерные объекты

5

В этой главе показан порядок использования объекта `WScript.Network` для управления сетевыми дисками и подключениями сетевых принтеров.

- Перед чтением этой главы следует изучить материал главы 3 «Создание сценариев и работа с объектами».
- В этой главе рассмотрено использование `WScript.Network` для управления печатью из DOS- и Windows-приложений.
- В последнем разделе показано, как можно отправить вывод из сценария непосредственно на принтер — это особенно важно для сценариев, запускаемых автоматически из Планировщика заданий (Task Scheduler).

Управление сетевыми и принтерными подключениями

Программы Windows Script Host (WSH) могут делать запросы к сетевым и принтерным подключениями вашего компьютера и управлять этими подключениями. Эти действия могут понадобиться по целому ряду причин:

- Сценарию, генерирующему документацию, может понадобиться собрать сетевую информацию и составить отчет.
- Может потребоваться сценарий входа в систему, гарантирующий при каждой регистрации ваших пользователей доступность нужных сетевых дисков и принтерных подключений, независимо от того, удачно или нет действовали пользователи во время своих предыдущих сеансов работы. (Хотя если используется Windows 7 или Vista и Windows Server 2008, Active Directory может позаботиться об этом за вас. Этот вопрос будет рассмотрен в данной главе чуть позже.)
- Ваше сетевое окружение подвержено частым изменениям, поэтому требуется воспользоваться сценарием для записи и восстановления правильных настроек.
- Требуется привести к единому стандарту и упростить настройку только что установленных в вашей организации компьютеров.

- Рабочая станция или сервер используются в качестве испытательного стенда, и требуется, не затрачивая особых усилий, выбрать одну из стандартных конфигураций.

WSH в состоянии помочь в выполнении всех этих задач. Хотя в режиме командной строки можно воспользоваться командой `net`, которая может выполнить ряд сходных задач, сценарии WSH предлагают более тонкое управление и могут работать не только с сетью, но и с реестром, файлами и другими системными компонентами.

Используемые в сценариях объекты позволяют также воспользоваться службами принтеров и (или) факсов для автоматического генерирования вывода из ваших сценариев. Печать и рассылка факсов из сценариев рассматриваются после рассмотрения вопросов управления сетями и принтерами в главе 6 «Объекты сообщений и факсов».

-
- ★ Следует также запомнить, что объекты, рассматриваемые в этих главах, можно использовать в макросах Microsoft Word и других сценарных приложениях, а не только в WSH.
-

Первым представляемым в данной главе объектом будет `WSHNetwork`, который создается с использованием имени `"WScript.Network"`, как показано в следующем операторе:

```
set wshNetwork = CreateObject("WScript.Network")
```

Это весьма практичный объект. Как и в случае использования ранее рассмотренного объекта `Scripting.FileSystemObject`, для выполнения сетевых задач в сценарии нужен будет только один экземпляр объекта `WScript.Network`. Его свойства и методы показаны в справке 5.1.

СПРАВКА 5.1. СВОЙСТВА И МЕТОДЫ ОБЪЕКТА WSHNETWORK

СВОЙСТВА

ComputerName

Возвращает сетевое имя компьютера, установленное во вкладке **Имя компьютера** (**Computer Name**) диалогового окна **Свойства системы** (**System Properties**). Это имя компьютера используется при работе в сети Windows. (Только чтение.)

UserDomain

Возвращает имя сетевого домена Windows для учетной записи того пользователя, который находится в системе в настоящий момент. Для входов в систему под учетными записями, не использующими домены или являющимися локальными, это имя совпадает с именем компьютера. (Только чтение.)

UserName

Возвращает имя, под которым вошел в систему текущий пользователь. (Только чтение.)

МЕТОДЫ

AddPrinterConnection *LocalName*, *RemoteName*[, *UpdateProfile*[, *UserName*, *Password*]]

Перенаправляет один из эмулируемых портов LPT-принтера, доступных MS-DOS-приложениям на локальном компьютере на удаленный сетевой принтер.

При этом используются следующие аргументы:

- *LocalName* — имя порта принтера DOS ("LPT1", "LPT2" или "LPT3"), используемое для перенаправления на сетевой принтер. Перенаправить LPT-порт, который уже используется физическим локальным принтером, невозможно.
- *RemoteName* — общее имя сетевого принтера, соответствующее универсальному соглашению по присваиванию имен — Universal Naming Convention (UNC) (*\\машина\общее_имя*). Можно указать укороченное общее имя принтера или его «дружеское» имя (например, "\\sumatra\okidata" или "\\sumatra\Okidata 0L 810").
- *UpdateProfile* — необязательное булево значение. Если аргумент имеет значение True, отображение принтера сохраняется в профиле текущего пользователя, чтобы имелась возможность его восстановления в процессе следующего входа в систему.
- *UserName* — необязательный аргумент. Это имя учетной записи пользователя, допустимое на удаленном компьютере для использования в данном сетевом подключении. Если уже есть подключение сетевого диска или принтера на том же удаленном компьютере, использующее имя пользователя по умолчанию или какое-нибудь другое имя пользователя, этот аргумент не работает. (Для подключения к конкретному удаленному компьютеру одновременно может использоваться только один набор авторизационных данных.) Аргумент может быть указан в виде *"имя_пользователя"* для использования одной из локальных учетных записей на удаленной машине или в виде *"\\домен\имя_пользователя"* для использования учетной записи домена.
- *Password* — необязательный аргумент. Пароль входа в систему, связанный с указанным именем пользователя, — *UserName*.

Если порт уже перенаправлен или используется подключенным к нему локальным принтером, генерируется ошибка.

AddWindowsPrinterConnection *PrinterPath* [, *DriveName* [, *Port*]]

Добавляет подключение к удаленному сетевому принтеру, который после этого доступен Windows-приложениям. При этом в пользовательскую папку **Принтеры** (Printers) добавляется значок.

Общее имя сетевого принтера в UNC-нотации (например, "*\\машина\общее_имя*") указывается в аргументе *PrinterPath*. Можно указать сокращенное общее имя принтера или его дружеское имя в том виде, как оно отображено удаленным компьютером.

Аргументы *DriverName* и *Port* требуются только когда сценарий запускается на компьютерах, работающих под управлением Windows 98 или Me, поэтому здесь они не рассматриваются. В сценариях их следует опускать.

Если значок подключения для указанного принтера уже существует, ошибка не генерируется.

EnumNetworkDrives()

Возвращает коллекцию объектов, перечисляющих буквы дисков, отображаемых для удаленных сетевых дисков. Объекты содержат строковые значения и используются попарно. Первый элемент, `.Item(0)`, является именем отображенного диска, например "F:". Второй элемент, `.Item(1)`, является общим UNC-именем, на которое осуществлено перенаправление диска. Точно так же `.Item(2)` является именем диска, а `.Item(3)` — соответствующим общим именем и т. д. Мы рассмотрим чуть позже в данном разделе, как работать с этим странным расположением.

EnumPrinterConnections()

Возвращает коллекцию объектов, в которых перечисляются LPT-порты, перенаправленные на удаленные сетевые принтеры. Как и в случае с `EnumNetworkDrives`, эти объекты содержат простые текстовые строки и используются попарно. Первый элемент каждой пары является локальным именем порта (например, "LPT1"). Второй элемент каждой пары является именем удаленного принтера. Возвращенные имена могут быть не короткими общими именами в формате UNC, а «дружественными» именами принтеров удаленной машины. Например, вместо `\\sumatra\okidata` имя может быть указано в списке как `\\sumatra\Okidata 0L 810`.

MapNetworkDrive LocalName, RemoteName [, UpdateProfile [, UserName, Password]]

Перенаправляет обращения с указаниями буквы локального диска к удаленной общей папке. При этом используются следующие аргументы:

- *LocalName* — буква диска, используемая для перенаправления. Этот аргумент может быть также пустой строкой для создания подключения к удаленному общему ресурсу без отображения на букву; это может увеличить скорость последовательного обращения к удаленному диску.
- *RemoteName* — общее имя удаленной папки в формате UNC (например, `\\сервер\общее_имя`).
- *UpdateProfile* — необязательный булев аргумент. Если этот аргумент передан методу и имеет значение `True`, отображение диска добавляется к профилю учетной записи текущего пользователя, чтобы оно восстанавливалось при следующем входе в систему.
- *UserName* — необязательный аргумент. Это имя учетной записи пользователя, допустимое на удаленном компьютере для использования в данном сетевом подключении. Если уже есть подключение сетевого диска или принтера на том же удаленном компьютере, использующее имя пользователя по умолчанию или какое-нибудь другое имя пользователя, этот аргумент не работает. (Для подключения к конкретному удаленному компьютеру одновременно может использоваться только один набор авторизационных данных.) Аргумент может быть указан в виде `"имя_пользователя"` для использования одной из локальных

учетных записей на удаленной машине или в виде "\\домен\имя_пользователя" для использования учетной записи домена.

- *Password* — необязательный аргумент. Это пароль для использования при аутентификации указанного имени пользователя — *UserName*.

Если буква диска уже задействована в отображении, буква соответствует локальному физическому диску или если сетевой путь не существует, этот метод дает сбой.

RemoveNetworkDrive Name [, Force [, UpdateProfile]]

Отключает отображение на букву диска удаленной общей папки.

При этом используются следующие аргументы:

- *Name* — буква отключаемого диска (например, "F:"). Учтите, что двоеточие является обязательным элементом. Если сетевой путь был подключен без указания буквы диска, вместо буквы следует указать именно этот путь.
- *Force* — необязательный булев аргумент. Если он указан и имеет значение True, отображение на букву диска отключается, даже если оно еще используется. В противном случае, если отображение используется, сценарий генерирует ошибку. Использование означает, что локальная программа использует файл на удаленном диске или что окно командной строки имеет данное отображение в качестве своего текущего диска.
- *UpdateProfile* — необязательный булев аргумент. Если этот аргумент указан и имеет значение True, отображение диска удаляется из профиля учетной записи текущего пользователя, чтобы оно не восстанавливалось автоматически при следующем входе в систему. Если отображение записано в профиль и аргумент *UpdateProfile* не указан или имеет значение False, отображение восстанавливается при следующем входе в систему.

Если подключение отсутствует, генерируется ошибка.

RemovePrinterConnection Name [, bForce [, bUpdateProfile]]

Удаляет перенаправление с локального эмулируемого LPT-порта на удаленный сетевой принтер.

При этом используются следующие аргументы:

- *Name* — имя перенаправленного порта (например, "LPT1") или общее имя в формате UNC, на которое был перенаправлен порт; последнее указание может удалить отображение нескольких LPT-портов на один и тот же принтер. Для отмены DOS-перенаправления укажите в *Name* имя локального перенаправленного DOS-порта (например, "LPT2").
- *Force* — необязательный булев аргумент. Если он указан и имеет значение True, перенаправление удаляется, даже если программа на локальном компьютере буферизует вывод на принтер.
- *UpdateProfile* — см. описание такого же аргумента для метода *RemoveNetworkDrive*.

Если указанное подключение или принтер отсутствует, генерируется ошибка.

SetDefaultPrinter *PrinterName*

Устанавливает общий сетевой принтер в качестве принтера по умолчанию на локальном компьютере. Аргумент *PrinterName* является общим именем удаленного принтера в формате UNC. Этот метод нужно применять после использования метода `AddWindowsPrinterConnection`, создающего подключение к тому же общему принтеру.

Этот метод не может выбрать в качестве используемого по умолчанию локальный принтер, а также не может предоставить какой-нибудь способ определить текущий, используемый по умолчанию принтер.

Следует учесть, что объект `WScript.Network` ограничен в своих возможностях. Он может выполнять следующие задачи:

- Допускать просмотр, но не изменение пользовательского имени, под которым был осуществлен вход в систему, имени компьютера и домена текущего входа в систему.
- Допускать составление списка, добавление и удаление отображения дисков, но не может обеспечивать совместный доступ к папкам или принтерам в сети.
- Допускать составление списка, добавление и удаление подключения к общим сетевым принтерам, но не может обеспечивать установку локальных (непосредственно подключенных) принтеров.
- Допускать установку в качестве одного из сетевых принтеров как принтера по умолчанию, но только если это не локальный принтер.

Для выполнения более сложных задач управления сетью нужно обратиться к материалам главы 7 «Инструментарий управления Windows» и главы 8, «Active Directory Scripting Interface». И все же, как будет показано в следующих нескольких разделах, `WScript.Network` полезен для выполнения многих базовых задач при работе с сетью.

Извлечение сетевой пользовательской информации

Три свойства объекта `WScript.Network`, как показано в следующем учебном сценарии, извлекают информацию о текущем пользователе, компьютере и домене, с которого пользователь вошел в систему:

```
set sn = CreateObject("WScript.network")
WScript.echo sn.userName
WScript.echo sn.computerName
WScript.echo sn.userDomain
```

На моем компьютере, к примеру, выводится следующая информация:

```
bknittel
JAVA
JAVA
```

Нетрудно догадаться, что мое пользовательское имя, под которым я вошел в систему, было `bknittel`, а используемый мной компьютер назывался `Java`. Свойство `UserDomain` также имело значение `Java`, даже при том, что оно не было именем моей рабочей группы. Почему? Свойство `UserDomain` сообщает о том, кто именно разрешил использование этой учетной записи. Это либо доменное имя, либо имя локального компьютера, поскольку учетные записи пользователей берутся из одного из двух мест:

- **локальных учетных записей**, устанавливаемых только на отдельном компьютере. Когда текущий пользователь входит в систему, используя локальную учетную запись, свойство `UserDomain` показывает имя компьютера;
- **учетных записей домена**, устанавливаемых администратором сетевого домена. Если это разрешено настройками сетевой безопасности, пользователи одного домена могут регистрироваться на компьютерах, являющихся частью другого домена. Поэтому информация, сообщаемая свойством `UserDomain`, может не обязательно быть именем домена, которому принадлежит компьютер, и, в любом случае, не совпадает с именем компьютера.

На компьютерах, входящих в рабочую группу, а не в сетевой домен, включая все компьютеры, работающие под управлением Windows Home Edition (которые не могут входить в домен), все учетные записи являются локальными, поэтому значение свойства `UserDomain` всегда соответствует имени компьютера.

Сам по себе вывод информации о пользователе и домене может показаться не слишком полезным. И тем не менее получение доступа к этой информации может пригодиться в сценариях, используемых более чем одним пользователем или более чем на одном компьютере. Создание сценария, который может быть запущен всеми пользователями, может облегчить жизнь сетевого администратора, но при этом может потребоваться обеспечить различные настройки на основе имени компьютера или пользователя. Для этого можно воспользоваться следующими способами:

- Можно написать сценарий, настраивающий отображение дисков, значения переменных среды окружения или содержимого переменной `PATH`, чтобы туда были включены папки, в названиях которых есть имя пользователя. Например, сеть может иметь наборы общих папок вида `\\СЕРВЕР\домашние_папки\имя_пользователя`, по одному для каждого пользователя, где пользователь может хранить в стандартном месте файлы, требующие ежедневного резервного копирования. Отображение буквы диска на такую «домашнюю» папку может быть автоматизировано в доменной сети, но вы-то сами находитесь в своей собственной сети рабочей группы. И тем не менее с этим можно справиться: в сценарии входа в систему можно воспользоваться свойством `UserName`, чтобы составить нужное имя общей папки и использовать его для отображения этой папки на букву диска.
- Можно написать сценарий, совершающий конкретные действия только для определенных пользователей или компьютеров. Например, после настройки стандартных программ и отображений можно воспользоваться сценарием входа в систему, запускающим конкретные программы только для того компьютера,

о котором известно, что у него имеется дополнительное оборудование или можно выполнить специальные настройки для определенных пользователей.

Рассмотрим пример, показывающий, как универсальный сценарий входа в систему может выполнять специальные операции для каждого пользователя:

```
' Файл примера script0501.vbs
set wshNetwork = CreateObject("WScript.Network")
set wshShell   = CreateObject("WScript.Shell")

user = ucase(wshNetwork.userName)

MapDrive "H:", "\\bali\home\" & user   ' отображение на букву диска папки home

select case user
  case "BKNITTEL"           ' отображения на буквы F и G
    MapDrive  "F:", "\\sumatra\photos"
    MapDrive  "G:", "\\ambon\software"
    wshShell.run "notepad c:\todo.txt" ' вывод списка заданий на день
    UsePrinter "\\bali\okidata"

    case "NALEKS"           ' отображение на букву диска папки симулятора
    MapDrive  "F:", "\\bali\ibm360"

    case "ACCOUNTING"
    MapDrive  "X:", "\\sumatra\quicken"

    case else               ' для любых других пользователей просто установка сетевого
    ' принтера
    UsePrinter "\\bali\hp laserjet"
end select

function MapDrive (byval drive, path)
...
```

Функции `MapDrive` и `UsePrinter` будут рассмотрены в данной главе чуть позже. Сценарии входа в систему еще будут рассматриваться в главе 9 «Использование сценариев для управления компьютерами и сетями».

Если в систему входит пользователь `bknittel`, этот сценарий создает отображения на две буквы диска, запускает программу Блокнот и гарантирует доступность указанного принтера; если же в систему входит пользователь `naleks`, создается другое отображение и т. д. Как видите, подобный сценарий может позволить вставить всю информацию по входу в систему в одно место. Если поместить ярлык этого сценария в папку автозагрузки для всех пользователей, он будет запускаться, когда в систему будет входить любой из пользователей. Я имею в виду папку `\Program Data\Microsoft\Windows\Start Menu\Programs\Startup` на Windows 7 и Vista и папку `\Documents and Settings\All Users\Start Menu\Programs\Startup` на Windows XP.

- ▼ Такая разновидность настройки под конкретного пользователя полезна как на небольших сетях рабочих групп, так и на больших корпоративных доменах. При большом количестве пользователей может быть полезнее выполнять настроечные операции на основе групповой принадлежности, а не на основе индивидуальных пользовательских имен. Как проверить принадлежность к группе, будет показано в главе 8.

При всей своей пользе для извлечения этих пользовательских и компьютерных имен `WScript.Network` не позволяет изменять эти данные или добывать информацию о пользовательских привилегиях или настройках учетных записей. Для этих целей нужно воспользоваться средствами Windows Management Instrumentation (WMI), которые рассматриваются в главе 7.

Управления отображениями на буквы дисков

Одной из наиболее востребованных сетевых функций, используемых в сценариях, является отображение общих сетевых папок на буквы дисков. Обычно такое отображение создается пользователем с помощью программы Explorer или путем использования в командной строке команды `net`. Отображение, созданное в отдельном сеансе входа в систему, обычно записывается в пользовательском профиле и становится доступным снова при следующем входе в систему. Но иногда требуется более конкретное управление отображением, например следующее:

- В сценарии входа в систему может потребоваться установить стандартное отображение на буквы дисков, в котором пользователи постоянно нуждаются. Сценарий должен переписать изменения, внесенные пользователями в предыдущем сеансе входа в систему, чтобы гарантировать доступность правильного отображения. (Лучше будет сказать: «Даже если пользователь все перепутает».)
- В сценариях, выполняющих полезные задачи, создающих резервные копии и обслуживающих систему, может понадобиться создавать временные отображения на буквы дисков, чтобы файлы были перемещены в другие сетевые места.
- Отображения на буквы дисков могут понадобиться в сценариях, запускающихся автоматически Планировщиком заданий (Task Scheduler). Но по умолчанию Планировщик заданий запускает сценарии в контексте специальной системной учетной записи пользователя, а не в контексте вашей собственной учетной записи. Поэтому ваше персональное стандартное отображение на буквы дисков недоступно. Сценарию понадобится создать все необходимые ему отображения.

Методы, позволяющие в подобных ситуациях управлять дисковым отображением, предоставляются объектом `WScript.Network`.

Создание перечня дисковых отображений с помощью `EnumNetworkDrives`

Свойство `EnumNetworkDrives` возвращает коллекцию объектов, описывающих все имеющиеся отображения на буквы дисков. Но эта коллекция не похожа на те, которые нам попадались до сих пор, когда коллекция содержала полноценные объекты, каждый из которых имел свои собственные свойства и методы. Эта коллекция относится к разряду тех, которые должны быть изучены путем просмотра ее свойств `Item` в порядке их нумерации. Коллекция представляет собой простой

перечень текстовых строк, которые, если взять их попарно, описывают текущее отображение.

→ Дополнительные сведения о коллекциях и свойстве `Item` даны в главе 3, в разделе «Контейнеры и коллекции».

Значения из данной коллекции могут быть извлечены путем обращения к свойству `Item`. Первая строка (`Item 0`) дает букву диска, на которую создано отображение (например, "F:"). Вторая строка (`Item 1`) дает общее имя, отображаемое на букву диска в виде пути в формате UNC. Коллекция таким образом продолжается парами строк — третья строка (`Item 2`) дает еще одну букву диска, на которую создано отображение и т. д. Например, сценарий

```
set wshNetwork = CreateObject("WScript.Network")
set maps = wshNetwork.EnumNetworkDrives
```

может выдать коллекцию, подобную той, что показана в листинге 5.1.

Листинг 5.1. Пример коллекции `EnumNetworkDrives`

```
maps.item(0)           "F:"           ' имя и путь
maps.item(1)           "\\sumatra\chapters" ' первого диска
maps.item(2)           "H:"           ' имя и путь
maps.item(3)           "\\bali\home\bknittel" ' второго диска
.
.
.
maps.item(maps.Length-2) "X:"           ' имя и путь
maps.item(maps.Length-1) "\\bali\incoming faxes" ' последнего диска
```

Для исследования данной коллекции можно воспользоваться предпочитаемой вами в языке сценариев версией цикла `for`. Следующий сценарий, написанный на VBScript, выводит благодаря использованию шага цикла сдвоенный перечень элементов отображения:

```
' Файл примера script0502.vbs
set wshNetwork = CreateObject("WScript.Network")
set maps = wshNetwork.EnumNetworkDrives
for i = 0 to maps.Length-2 step 2
    WScript.echo "Диск", maps.item(i), "отображается на ", maps.item(i+1)
next
```

При каждом проходе цикла выражение `maps.item(i)` вычисляется в значение буквы диска, а выражение `maps.item(i+1)` вычисляется в значение пути к общей папке. На моем компьютере этот сценарий выдал следующую информацию:

```
Диск F: отображается на \\sumatra\chapters
Диск H: отображается на \\bali\home\bknittel
Диск J: отображается на \\bali\shared documents
Диск X: отображается на \\bali\incoming faxes
```

Вывод перечня отображенных дисков — далеко не самое полезное занятие, поскольку набор в окне командной строки Windows команды `net use` приведет к точно такому же результату. Но этот прием может пригодиться при отладке сценариев, запускаемых из Планировщика заданий. Используя инструментарий, позволяющий записывать данные в файлы, который был рассмотрен в главе 4 «Доступ к файлам